
 1

EURISOL DS/TASK12/TN-05-01 
1 June 2005 

The EURISOL Beta-beam Facility 

Parameter and Intensity Values, Version 1, April 2005 
 

M. Benedikt, S. Hancock, M. Lindroos 
 
 

Abstract 
The flux out of the Beta-beam facility [1] is determined by the number of ions that can 
be produced, by the number remaining after acceleration and by the total accumulated in 
the decay ring.  We present a “bottom-up” analysis of the ion intensities along the 
accelerator chain and of the neutrino and antineutrino flux out of the decay ring, starting 
with the rate at which atoms are transported out of the target. 
 

Beta-beam baseline 
A full description of the Beta-beam can be found in e.g. [2,3].  Here, we only repeat the 
main elements of the baseline scenario, with emphasis on the actual parameters used for 
the intensity calculations.  This note documents intensities and certain parameters for 
the EURISOL Beta-beam conceptual design as it stands at the time of the first Task 
Meeting (14 April 2005).  The design will evolve with the objective of reaching the 
target values set by the user community.  All such developments will be documented in 
future EURISOL DS Beta-beam notes. 

Production 
In the Beta-beam baseline it is assumed that we can produce and transport 2×1013 6He 
atoms/s and 8×1011 18Ne atoms/s out of the target with a proton driver beam current of 
100 μA impinging on the target at 2.2 GeV.  These numbers are based on an evaluation 
of suitable isotopes and their production rates at ISOL facilities and have been presented 
as part of the Beta-beam contribution to NuFact02.  The full write-up of this 
presentation could not be published as part of the proceedings, but it exists as a CERN 
internal note [4]. 

ECR source 
Efficient bunching and stripping of the high intensity beam are achieved using a high-
frequency ECR source [5].  We assume the efficiency of such a source to be 100% for 
6He, with all ions being extracted in the 2+ charge state, but only 30% for 18Ne ions due 
to only one (e.g., 6+) of several charge states being extracted.  We further assume that 
the ECR operates at 16 Hz with an accumulation time of 60 ms, that the source is at a 
potential of 50 kV relative to the linac, and that the ions are ejected in 50 μs long 
bunches with a physical transverse emittance of around 50 π.mm.mrad. 

Linac and rapid cycling synchrotron 
The linac accelerates the ions to 100 MeV/nucleon, after which they are multi-turn 
injected into a rapid cycling synchrotron (RCS).  We assume an injection efficiency of 

50%.  Both ion species are accelerated to a magnetic rigidity of 8 Tm in a time dictated 
by the 16 Hz repetition rate of the ECR source.  Each cycle provides a single bunch. 

PS accumulation and acceleration to high energy 
The PS waits for 16 shots from the RCS then accelerates the 16 bunches to a magnetic 
rigidity of 86.7 Tm.  Merging takes place at top energy to form 8 bunches that are 
ejected in a single turn to the SPS.  There they are accelerated to γ=100 and ejected, 
again by single-turn extraction, into the decay ring.  The cycle time of the SPS is a 
multiple of the 1.2 s basic period of the CERN machines.  The extra acceleration 
required for the 6He results in a 6 s cycle compared with only 3.6 s for 18Ne. 

Decay ring 
The 8 incoming bunches are combined with the 8 circulating ones of the stack by an 
asymmetric merging process in the decay ring.  The number of times this process can be 
repeated is constrained by the 1 eVs longitudinal emittance of each bunch delivered by 
the SPS.  Simulations for 6He show that such bunches can be stacked up to 15 times.  
The situation should be better for 18Ne due a more advantageous charge-to-mass ratio, 
but the necessary simulations have not yet been performed and, therefore, the same 
merging factor is conservatively assumed. 
 
The total duration (at its shortest) of the 8 bunches in the 6He simulations is roughly 
50 ns.  Since the revolution period of the decay ring is 23 μs, this corresponds to a so-
called duty cycle of around 2×10-3. 
 
The annual integrated flux of potentially useful neutrinos and antineutrinos emanating 
from the decay ring is linearly dependent on the relative length of the straight section 
that points towards the detector.  This is taken to be 36% of the decay ring 
circumference [6]. 

Results 
The parameters given in the Appendix to this note result in the following numbers of 
ions at each stage of the CERN Beta-beam facility.  The decay losses are properly 
accounted for, but the transfer efficiencies between the different machines are assumed 
to be 100% except for the multi-turn injection into the RCS.  The source rate is given (in 
atoms/s) at the entrance to the ECR, while the decay ring figure is the maximum number 
of stored ions immediately after injection from the SPS. 
 
6He: 
 
Source rate 2. × 1013

ECR 1.17 × 1012

RCS inj 5.83 × 1011

RCS 5.71 × 1011

PS inj 6.85 × 1012

PS 5.8 × 1012

SPS 5.48 × 1012

Decay Ring 5.88 × 1013  

18Ne: 
 
Source rate 8. × 1011

ECR 1.42 × 1010

RCS inj 7.1 × 109

RCS 7.04 × 109

PS inj 1.01 × 1011

PS 9.58 × 1010

SPS 9.45 × 1010

Decay Ring 1.19 × 1012  
 
This corresponds to a total integrated flux in a ten-year run (5 years of 6He plus 5 years 
of 18Ne) of 8.82×1018 antineutrinos and 9.49×1016 neutrinos. 
 
There is little consensus on target values for the integrated flux out of the decay ring, 
nor is γ=100 for both ion species well-established.  However, the desired order of 
magnitude is 1018 decays in a canonical year of 107 s [7].  (We do not consider running 
with 6He and 18Ne ions simultaneously.) 

Conclusions 
The integrated fluxes achievable in the present scenario have been evaluated in a 
consistent “bottom-up” approach.  The results for neutrinos coming from the decay of 
18Ne do not match the expected order of magnitude. 

Appendix 
 
Mathematica package in text form: 
 
 
Off[General::spell1]; 
 
c = 299792458; 
 
(* All energies are in eV; "Tpern" = kinetic energy per nucleon. *) 
setmachines := ( 
  bp = 1.2; (* 1.2s basic period. *) 
  quantize[t_] := bp (IntegerPart[t/bp] + 
Ceiling[FractionalPart[t/bp]]); 
  ecrejTpern := 50 10^3 q/nnumber; 
  ecraccumulationtime = 60 10^-3; 
  rcsinjTpern = 100 10^6; 
  rcsaccumulationtime = 50 10^-6; 
  rcsefficiency = 0.5; 
  bunchingtime = 5 10^-3; 
  rcsaccelerationtime := (rcscycletime - bunchingtime)/2; 
  rcsBrhomax = 8; (* 300MeV/nucleon 6He equivalent. *) 
  rcscycletime = 1/16; 
  rcsbatches = 16; 
  psinjTpern := Sqrt[(rcsBrhomax c q/nnumber)^2 + Epern^2] - Epern; 
  psaccumulationtime := (rcsbatches-1) rcscycletime; 
  psaccelerationtime = 0.8; 
  psBrhomax = 86.7; 
  pscycletime := quantize[psaccumulationtime + 2.0 psaccelerationtime 
+ 0.1]; 
  psbatches = 1; 
  spsinjTpern := Sqrt[(psBrhomax c q/nnumber)^2 + Epern^2] - Epern; 
  spsaccumulationtime := (psbatches-1) pscycletime; 
  spsaccelerationtime := (topTpern - spsinjTpern) * 
    nnumber/(q 100 10^9); (* 100GeV/s proton equivalent. *) 
  topgamma = 100; 
  topTpern := Epern (topgamma - 1); 
  spscycletime := quantize[spsaccumulationtime + 1.5 
spsaccelerationtime + 1.0]; 
  spsrepetitiontime := Max[pscycletime, spscycletime]; 
  straightfraction = 0.36; 
) 
 
set6He := ( 
  name = "6He"; nnumber = 6; q = 2; 
  Erest = 5.606 10^9; (* Rest mass in eV/c^2. *) 
  Epern := Erest/nnumber; 
  thalf = 0.81; (* Half-life at rest. *) 
  sourcerate = 2 10^13; (* Effective rate, includes 40% efficiency. *) 
  ecrefficiency = 1; 
  setmachines; 
  mergesratio = 15; 
) 
 
set18Ne := ( 
  name = "18Ne"; nnumber = 18; q = 10; 
  Erest = 16.767 10^9; (* Rest mass in eV/c^2. *) 
  Epern := Erest/nnumber; 
  thalf = 1.67; (* Half-life at rest. *) 

  sourcerate = 8 10^11; (* Effective rate, includes 40% efficiency. *) 
  ecrefficiency = 0.3; (* Single charge state. *) 
  setmachines; 
  mergesratio = 15; 
) 
 
ecraccumulation := (ClearAll[n]; 
  gamma[t_] := 1 + ecrejTpern / Epern; 
  decayrate[t_] := Log[2] n[t] / (gamma[t] thalf); 
  eqns = {D[n[t], t] == sourcerate - decayrate[t], n[0]==0}; 
  n[t_] = ecrefficiency n[t] /. DSolve[eqns, n[t], t] //First; 
  nout0 = n[ecraccumulationtime] 
) 
 
rcsbunching := (ClearAll[n]; 
  gamma[t_] := 1 + rcsinjTpern / Epern; 
  decayrate[t_] := Log[2] n[t] / (gamma[t] thalf); 
  eqns = {D[n[t], t] == -decayrate[t], n[0]==nout0}; 
  n[t_] = rcsefficiency n[t] /. DSolve[eqns, n[t], t] //First; 
  nout1 = n[bunchingtime] 
) 
 
rcsacceleration := (ClearAll[n]; 
  rcsTpern[t_] := rcsinjTpern + 
    (psinjTpern - rcsinjTpern) (1 - Cos[Pi t/rcsaccelerationtime])/2; 
  gamma[t_] := 1 + rcsTpern[t] / Epern; 
  decayrate[t_] := Log[2] n[t] / (gamma[t] thalf); 
  eqns = {D[n[t], t] == -decayrate[t], n[0]==nout1}; 
  n[t_] = n[t] /. DSolve[eqns, n[t], t] //First; 
  nout2 = n[rcsaccelerationtime] 
) 
rcsout := (ecraccumulation; rcsbunching; rcsacceleration) 
 
psaccumulation := (ClearAll[n]; 
  gamma[t_] := 1 + psinjTpern / Epern; 
  decayrate[t_] := Log[2] n[t] / (gamma[t] thalf); 
  eqns = {D[n[t], t] == -decayrate[t], n[0]==nout2}; 
  nsinglebatch[t_] = n[t] /. DSolve[eqns, n[t], t] //First; 
  n[t_] = Sum[UnitStep[t-t0] nsinglebatch[t-t0], 
    {t0, 0,psaccumulationtime,rcscycletime}]; 
  nout3 = n[psaccumulationtime] 
) 
 
psacceleration := (ClearAll[n]; 
  psTpern[t_] := psinjTpern + 
    (spsinjTpern - psinjTpern) t/psaccelerationtime; 
  gamma[t_] := 1 + psTpern[t] / Epern; 
  decayrate[t_] := Log[2] n[t] / (gamma[t] thalf); 
  eqns = {D[n[t], t] == -decayrate[t], n[0]==nout3}; 
  n[t_] = n[t] /. DSolve[eqns, n[t], t] //First; 
  nout4 = n[psaccelerationtime] 
) 
psout := (rcsout; psaccumulation; psacceleration) 
 
spsaccumulation := (ClearAll[n]; 
  nout5 = nout4 
) 
 
spsacceleration := (ClearAll[n]; 
  spsTpern[t_] := spsinjTpern + 
    (topTpern - spsinjTpern) t/spsaccelerationtime; 

  gamma[t_] := 1 + spsTpern[t] / Epern; 
  decayrate[t_] := Log[2] n[t] / (gamma[t] thalf); 
  eqns = {D[n[t], t] == -decayrate[t], n[0]==nout5}; 
  n[t_] = n[t] /. DSolve[eqns, n[t], t] //First; 
  nout6 = n[spsaccelerationtime] 
) 
spsout := (psout; spsaccumulation; spsacceleration) 
 
decayringaccumulation := (ClearAll[n]; 
  gamma[t_] := 1 + topTpern / Epern; 
  decayrate[t_] := Log[2] n[t] / (gamma[t] thalf); 
  eqns = {D[n[t], t] == -decayrate[t], n[0]==nout6}; 
  nsinglebatch[t_] = n[t] /. DSolve[eqns, n[t], t] //First; 
  n[t_] = Sum[UnitStep[t-t0] 
UnitStep[t0+mergesratio*spsrepetitiontime-t] * 
    nsinglebatch[t-t0], {t0, 0,t,spsrepetitiontime}]; 
  nout7 = n[(mergesratio-1)spsrepetitiontime] 
) 
fullchain := (spsout; decayringaccumulation) 
 
annualrate := (1 - 2^-(mergesratio spsrepetitiontime/(topgamma 
thalf))) * 
  spsout straightfraction 10^7/spsrepetitiontime 


